Simulation **P**rogram for **I**ntegrated **C**ircuits **E**mphasis

Capacidade de análise

- **Non-linear DC analysis: calculates the DC transfer curve.**
- **Non-linear transient analysis: calculates the voltage and current as a function of time when a large signal is applied.**
- **Linear AC Analysis: calculates the output as a function of frequency. A bode plot is generated.**
- Noise analysis.
- Distortion analysis.
- **Fourier analysis: calculates and plots the frequency spectrum.**
- Monte Carlo Analysis.
- Analog and Digital libraries of standard components (such as NAND, NOR, flip-flops, and other digital gates, op amps, etc)

Componentes

- Independent and dependent voltage and current sources
- Resistors
- Capacitors
- Inductors
- Mutual inductors
- Operational amplifiers
- Switches
- Diodes
- Bipolar transistors
- MOS transistors: JFET; MOSFET

Características

- Data statements: description of the components and the interconnections.
- Control statements: tells SPICE what type of analysis to perform on the circuit.
- Output statements: specifies what outputs are to be printed or plotted.

```
TITLE STATEMENT
ELEMENT STATEMENTS
.
.
COMMAND (CONTROL) STATEMENTS
OUTPUT STATEMENTS
.END <CR>
```

Características

- Numbers can be integers, or floating points:

```
RES1 1 0 3500

or

RES1 1 0 3.5E3
```

- One can also use the following scale factors:

```
T(= 1E12 or 10+12); G(= E9); MEG(= E6); K(= E3); M(= E-3); U(= E-6);
N(= E-9); P(= E-12), and F(= E-15)
```

- Both upper and lower case letters are allowed in Pspice:

```
225P, 225p, 225pF; 225pFarad; 225E-12; 0.225N, etc.
```

Especificação da topologia do circuito

Independent DC Sources:

Voltage source: `Vname N1 N2 Type Value`
Current source: `Iname N1 N2 Type Value`

`N1` is the positive terminal node
`N2` is the negative terminal node





Type can be `DC`, `AC` or `TRAN`, depending on the type of analysis
Value gives the value of the source

The name of a voltage and current source must start with `V` and `I`, respectively.
Examples:

```
Vin 2 0 DC 10
Is 3 4 DC 1.5
```

Especificação da topologia do circuito

Resistors:

```
Rname N1 N2 Value
```

Capacitors (C) and Inductors (L):

```
Cname N1 N2 Value <IC>
Lname N1 N2 Value <IC>
```

`N1` is the positive node.
`N2` is the negative node.
`IC` is the initial condition (DC voltage or current).

The symbol < > means that the field is optional. If not specified, it is assumed to be zero. In case of an inductor, the current flows from N1 to N2. Example:

```
Cap5 3 4 35E-12 5
L12 7 3 6.25E-3 1m
```



Especificação da topologia do circuito

Sinusoidal sources

```
Vname N1 N2 SIN(VO VA FREQ TD THETA PHASE)
```

which is a damped sinusoidal voltage source:
Vname = VO + VA exp[-THETA.(t - TD)] sin[2pi.f (t - TD) + (Phase/360)]

`VO` - offset voltage in Volt.
`VA` - amplitude in volt.
`f` = `FREQ` - frequency in Herz.
`TD` - delay in seconds
`THETA` - damping factor per second
`Phase` - phase in degrees

If `TD`, `THETA` and `PHASE` are not specified, it is assumed to be zero.

Especificação da topologia do circuito

Sinusoidal sources

```
Vname N1 N2 SIN(VO VA FREQ TD THETA PHASE)
```

Example:

```
VG 1 2 SIN(5 10 50 0.2 0.1)
VG2 3 4 SIN(0 10 50)
```

The last example is an undamped, undelayed sinusoid with an amplitude of 10V and frequency of 50 Hz.

Especificação da topologia do circuito

Piecewise linear source (PWL)

```
Vname N1 N2 PWL(T1 V1 T2 V2 T3 V3 ...)
```

in which (Ti Vi) specifies the value Vi of the source at time Ti

Example:

```
Vgpwl 1 2 PWL(0 0 10U 5 100U 5 110U 0)
```



Especificação da topologia do circuito

Pulse

`Vname N1 N2 PULSE(V1 V2 TD Tr Tf PW Period)`

`V1` - initial voltage;
`V2` - peak voltage;
`TD` - initial delay time;
`Tr` - rise time;
`Tf` - fall time;
`PW` - pulse-width;
`Period` - period.



Especificação da topologia do circuito

Voltage Controlled

```
Sname N1 N2 C1 C2 Mname
```

Current Controlled

```
Wname N1 N2 Vname Mname
```

`N1` and `N2` are the terminals of the switch.
`C1` and `C2` are the controlling terminals.

`Vname` is the zero-value voltage source whose current controls the switch.
`Mname` refers to the switch model that is defined in another statement.

Especificação da topologia do circuito

```
.MODEL Mname Dname(Pvalues)
```

`Mname` refers to the `Mname` in the data statement of the switch.
`Dname` is the device name: `Vswitch` or `Iswitch`.
`Pvalues` are the values of the switch parameters, such as RON (on - resistance), ROFF (off-resistance), VON (the control voltage for ON; default = 1V), VOFF (control voltage for OFF; default = 0V), etc.

Example:

```
Sname N1 N2 C1 C2 Mname
```

```
S15 3 5 8 9 SMOD
.MODEL SMOD VSWITCH(RON = 10, VON = 0, ROFF = 100MEG)
```

Especificação da topologia do circuito

Subcircuits

A subcircuit allows you to define a collection of elements as a subcircuit (e.g. an operational amplifier) and to insert this description into the overall circuit (as you would do for any other element).

Defining a subcircuit

A subcircuit is defined bu a .SUBCKT control statement, followed by the circuit description as follows:

```
.SUBCKT SUBNAME N1 N2 N3 ...
Element statements
.
.
.
.ENDS SUBNAME
```

Especificação da topologia do circuito

Using a subcircuit:

```
Xname N1 N2 N3 ... SUBNAME
```

`Xname` refers to the element (subcircuit) being used

Example:

```
vs 1 0 dc 5
r1 1 2 200
rf 2 3 1k
x1 0 2 3 opamp741
.dc vs 0 10 1
.plot dc v(3)
.end
```

Especificação da topologia do circuito

Semiconductor Devices

```
.MODEL MODName Type (parameter values)
```

`MODName` is the name of the model for the device.

`Type` refers to the type of device and can be any of the following:

D: Diode
NPN: NPN bipolar transistor
PNP: PNP bipolar transistor
NMOS: NMOS transistor
PMOS: PMOS transistor
NJF: N-channel JFET model
PJF: P-channel JFET model

Especificação da topologia do circuito

Diode

Element line: `Dname N+ N− MODName`

The element name starts with `D` to indicate that the element is a diode, `N+` and `N−` are the two node numbers

Model statement:

```
.MODEL MODName D (IS= N= Rs= CJO= Tt= BV= IBV=)
```

As an example, the model parameters for a 1N4148 commercial diode are as follows:

```
.model D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA)
```

Especificação da topologia do circuito

Bipolar transistors

Element: `Qname C B E BJT_modelName`

Model statement:

```
.MODEL BJT_modName NPN (BF=val IS=val VAF=val ...)
```

`BF` is the common emitter current gain ß,
`IS` is the saturation current and
`VAF` is the Early voltage.
Default values are assumed (ß=100; IS=1E-16A, and VAF=[infinite]). As an example, the model parameters for the 2N2222A NPN transistor is given below:

```
.model Q2N2222A NPN (IS=14.34F XTI=3 EG=1.11 VAF= 74.03 BF=255.9 )
```

Comandos e Declarações de Análise

OP Statement

This statement instructs Spice to compute the DC operating points:

- Voltage at the nodes
- Current in each voltage source
- Operating point for each element

In PSpice it is usually NOT necessary to specify `.OP` as it gives you automatically the DC node voltages.

Comandos e Declarações de Análise

DC Statement

This statement allows you to increment (sweep) an independent source over a certain range with a specified step. The format is as follows:

```
.DC SRCname START STOP STEP
```

in which `SRC` name is the name of the source you want to vary; `START` and `STOP` are the starting and ending value, respectively; and `STEP` is the size of the increment.

Example: `.DC V1 0 20 2`

Comandos e Declarações de Análise

DC Statement

Nesting the DC sweep command

```
.DC SRCname1 START STOP STEP SRCname2 START STOP STEP
```

Example: `.DC Vds 0 5 0.5 Vgs 0 5 1`

In the example above, the voltage `Vds` will be swept from 0 to 5V in steps of 0.5V for every value of `Vgs (step of 1V).`

Comandos e Declarações de Análise

TF Statement

The `.TF` statement instructs PSpice to calculate the following *small signal* characteristics:

- the ratio of output variable to input variable (gain or tranfer gain)
- the resistance with respect to the input source
- the resistance with respect to the output terminals

```
.TF OUTVAR INSRC
```

in which `OUTVAR` is the name of the output variable and `INSRC` is the input source.

Example: `.TF V(3,0) VIN`

Comandos e Declarações de Análise

TRAN Statement

This statement specifies the time interval over which the transient analysis takes place, and the time increments. The format is as follows:

```
.TRAN TSTEP TSTOP <TSTART>
```

`TSTEP` is the printing increment.
`TSTOP` is the final time
`TSTART` is the starting time (if omitted, `TSTART` is assumed to be zero)

Comandos e Declarações de Análise

AC Statement

This statement is used to specify the frequency (AC) analysis. The format is as follows:

```
.AC LIN NP FSTART FSTOP

.AC DEC ND FSTART FSTOP

.AC OCT NO FSTART FSTOP
```

`LIN` stands for a linear frequency variation
`DEC` and `OCT` for a decade and octave variation respectively
`NP` stands for the number of points and `ND` and `NO` for the number of frequency points per decade and octave.
`FSTART` and `FSTOP` are the start and stopping frequencies in Herz

Example: `.AC DEC 10 1000 1E6`

Comandos de Saída

- These statements will instruct PSpice what **output** to generate.
- If you do not specify an output statement, PSpice will always calculate the DC operating points.
- The two types of outputs are the **prints** and **plots**.
- A **print** is a **table** of data points.
- A **plot** is a **graphical** representation.

The format is as follows:

```
.PRINT TYPE OV1 OV2 OV3 ...
.PLOT TYPE OV1 OV2 OV3 ...
```

Comandos de Saída

The format is as follows:

```
.PRINT TYPE OV1 OV2 OV3 ...
.PLOT TYPE OV1 OV2 OV3 ...
```

in which `TYPE` specifies the type of analysis to be printed or plotted and can be:

```
DC
TRAN
AC
```

Comandos de Saída

The output variables are `OV1,OV2, etc. and` can be voltage or currents in voltage sources.

Node voltages and device currents can be specified as magnitude (M), phase (P), real (R) or imaginary (I) parts by adding the suffix to V or I as follows:

`M`: Magnitude
`DB`: Magnitude in dB (deciBells)
`P`: Phase
`R`: Real part
`I`: Imaginary part

Examples:

```
.PLOT DC V(1,2) V(3) I(Vmeas)
.PRINT TRAN V(3,1) I(Vmeas)
.PLOT AC VM(3,0) VDB(4,2) VM(2,1) VP(3,1) IR(V2)
```

Exercícios

```
IBIAS  13  0  2.3mA
IAC    2   3  AC .001
IPULSE 1   0  PULSE(-1mA 1mA 2ns 2ns 2ns 50ns 100ns)
I3    26  77  SIN(.002 .002 1.5MEG)
```

Exercícios



Exercícios




```
* source EXEMPLO
C_C1          2 0  1n IC=0 TC=0,0
R_R1          1 2  1k TC=0,0
V_V1          1 0  5Vdc

*Analysis directives:
.DC LIN V_V1 0 5 1
```

Exercícios




```
* source EXEMPLO
C_C1         2 0  1n IC=0 TC=0,0
R_R1         1 2  1k TC=0,0
V_V1         1 0  5Vdc

*Analysis directives:
.DC LIN V_V1 0 5 1
```

V_V1	V(R1:2)
0	0
1	1
2	2
3	3
4	4
5	5

Exercícios

```
* source EXEMPLO 1
C_C1   2      0      1n  TC=0,0
R_R1   1      2      1k TC=0,0
V_V1   1      0      PWL(0 0 10u 0 10.001u 5 20u 5 20.001u 0)


.TRAN 200ns 30us
.PLOT TRAN  V[1]
.PLOT TRAN  V[2]
```

Exercícios

```
* source EXEMPLO 2
D_D1          1      2  D1N4148
.model D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA)

R_R1          2      0      1k TC=0,0
V_V1          1      0      SIN(0 10 60)
```

Exercícios

```
* source EXEMPLO 2
D_D1          1      2  D1N4148
.model D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA)

R_R1          2      0      1k TC=0,0
V_V1          1      0      SIN(0 10 60)

.TRAN 500us 32ms
.PLOT TRAN  V[2]
.PLOT TRAN  V[1,2]
```

Exercícios

```
* source EXEMPLO 3
R_R1   1      3      1992 TC=0,0
R_R2   3      0      300 TC=0,0
R_Rc   1      2      1600 TC=0,0
R_Re   4      0      390 TC=0,0
Q_Q1   2      3      4      Q2N2222A
.model Q2N2222A NPN (IS=14.34F XTI=3 EG=1.11 VAF= 74.03 BF= 87)

V_V1   1      0      DC 10V
```

Exercícios

```
* source EXEMPLO 3
R_R1   1      3      1992 TC=0,0
R_R2   3      0      300 TC=0,0
R_Rc   1      2      1600 TC=0,0
R_Re   4      0      390 TC=0,0
Q_Q1   2      3      4      Q2N2222A
.model Q2N2222A NPN (IS=14.34F XTI=3 EG=1.11 VAF= 74.03 BF= 87)

V_V1   1      0      DC 10V

.DC    V_V1   0      20     2

.PRINT        DC     V[2,4] V[3,4]
```

Exercícios

```
* source EXEMPLO 3
R_R1   1      3      1992 TC=0,0
R_R2   3      0      300 TC=0,0
R_Rc   1      2      1600 TC=0,0
R_Re   4      0      390 TC=0,0
Q_Q1   2      3      4      Q2N2222A
.model Q2N2222A NPN (IS=14.34F XTI=3 EG=1.11 VAF= 74.03 BF= 87)

V_V1   1      0      DC 10V
V_Vs   3      0      SIN(0.7 0.05 1k)

.TRAN 25us 1ms

.PLOT TRAN  V[3]
.PLOT TRAN  V[2]
```